
ABSTRACT 
 
The SAS® Output Delivery System (ODS) allows output to be printed directly to a Rich Text Format (RTF) file.  
RTF files can be viewed by Microsoft Word and other word-processing packages.  A table of contents (TOC) can 
greatly enhance an RTF file and make it more user-friendly.  Headers and footers, as well as custom footnotes, 
can also greatly enhance an RTF file.  This paper will describe a method to imbed “invisible” RTF control words 
into a file for the purpose of creating a custom TOC.  The example presented will also include document headers, 
footers, and custom footnotes.  The example code was run with SAS Version 9.1.2 on Windows XP Professional.  
The RTF file is viewed with Microsoft® Office Word 2003.  
 
 
INTRODUCTION 
 
The purpose of this paper is to describe a technique to include a customized table of contents (TOC) in a Rich 
Text Format (RTF) file created by SAS.  The example presented in this paper creates an RTF file viewed by 
Microsoft Word.  The first programming task is that the TOC must be reproducible (i.e. created by code).  The 
second programming task is that the RTF file must also include Microsoft Word headers and footers, and custom 
table titles and footnotes.   
 
Beginning with SAS version 8.1, RTF control words can be used within cells.  The technique presented here takes 
advantage of this feature; RTF control words for TOC entries are imbedded into the RTF document.  RTF controls 
words for formatting are also used to make the TOC marks “invisible”. 
 
In the example that follows, an RTF file will be created that contains a custom TOC, Microsoft Word headers and 
footers, and custom table titles and footnotes.  The complete SAS code can be found in Appendix I.     
 
 
BACKGROUND 
 
Table of Contents in Microsoft Word 
 
In Microsoft Word there are three common methods to create a table of contents: 

1) Create a table of contents from outline levels. 

2) Create a table of contents from custom styles. 

3) Create a table of contents from entries you mark yourself. 

Method 3 gives you the most flexibility in creating a custom TOC.  If you were to create a TOC in Word with 
method 3, you would use the Mark Table of Contents box to insert TOC fields into your document.  You choose 
exactly where the TOC marks go and choose the text that will appear in the TOC.  One very important limitation to 
this method is the inability to mark text that appears in a Word header or footer.  Another limitation is that this is a 
“manual” process and would need to be repeated whenever the RTF file was re-created. 
 

Table of Contents with SAS Code 
There are also three common methods to create a TOC using SAS code: 

1) Use the ODS RTF CONTENTS=YES option and the ODS PROCLABEL statement. 

2) Use the ODS WORDSTYLE option (Experimental in version 9). 

3) Imbed RTF control words for TOC specifications. 

Papers describing methods 1 and 2 can be found on the SAS Support website 
(http://support.sas.com/resources/papers/index.html).  Method 3 gives you the most flexibility, and will be the 
method used in this paper.  The technique will be to imbed RTF control words to mark TOC entries, and imbed 
RTF control words to create the actual TOC.  This will be similar to using Word Method 3 Create a table of 
contents from entries you mark yourself.   However, since the marks will be created by code, the TOC will be 
reproduced whenever the RTF file is recreated.    

 
Viewing Headers and Footers in Microsoft Word 
 
When ODS creates an RTF file, TITLE statements are converted to Word headers and FOOTNOTE statements 
are converted to Word footers.  If you would prefer that SAS titles and SAS footnotes appear as part of the body 
of the document you can use the BODYTITLE option.  The code would be: 
 ods rtf file = "the path name goes here/filename.RTF" bodytitle; 
 
In the example presented here, we want Word headers and footers, so will allow ODS to convert the SAS titles 
and footnotes. 
 
 
In-Line Formatting 
 
Beginning in SAS 8.2, ODS supports the ability to insert simple formatting text into a cell.  To use in-line 
formatting, you first define an escape character with the ODS statement: ODS ESCAPECHAR = ‘escape-
character’;.  You should choose an escape character that is not used in code for any other reason.  The ‘^’ 
character is usually a safe choice and will be used in this paper.   

ods escapechar = '^'; 
 
 
Custom Text with ODS RTF TEXT=  Statement 
 
The ODS RTF TEXT= statement allows text to be added between procedures.  The following code uses in-line 
formatting to modify the style of the cell. The outputwidth=100% option will create a cell that spans the width of 
the page, making the cell the same width as the page width.  The just=l option will specify left justification for the 
text.  This code will be used to create the custom table titles and footnotes in the RTF file.      
   ods rtf text="^S={outputwidth=100% just=l} Text Goes Here"; 
 
Since the ODS RTF TEXT= statement creates a cell in the RTF document, it can also be used to imbed RTF 
control words.  This is how the control words for the TOC will be imbedded.  RTF control words for formatting will 
also be imbedded to make the cell “invisible”.    

ods rtf text="^S={outputwidth=100% just=c} RTF Control Words Go Here"; 
 
 
 
METHODS 
 
Creating an RTF file in SAS 
 
The ODS code to create a basic RTF file is very simple.  The code is: 
 

ods rtf file = "the path name goes here/filename.RTF";  
       /* SAS procedures go here */ 
       run; 
       quit; 

ods rtf close; 
 

ODS RTF FILE = opens the destination and file; ODS RTF CLOSE closes the destination and file.  The RUN and 
QUIT commands before the ODS RTF close statement are required.  If QUIT is omitted, the output from the last 
procedure may not be included in the document.   

Setup for Example RTF File 
 
The following code will create a basic RTF file with one table of information (created by PROC PRINT).  The 
macro variables textformat_left and textformat_center will define the formats to use in the TITLE and FOOTNOTE 
statements.  ODS will convert the SAS TITLE and FOOTNOTE statements to Microsoft Word headers and 
footers.  The escape character is defined for in-line formatting used in later code.  The startpage=no option of the 
ODS RTF FILE= statement is used to prevent SAS from printing each procedure on a new page (the default).  In 
the final RTF file, two pages contain two tables each (created by PROC FREQ and PROC MEANS).  We will 
control when page breaks are made with the startpage= options; options include yes|no|now.  The notoc_data 
option is used to prevent SAS from adding default TOC commands to the document; we want only the TOC 
entries we create to appear in the final TOC. 
  

title;  
footnote; 

 
options orientation=landscape nodate center pageno=1; 

 
%let textformat_left   = font=Arial height = 12 pt justify = l; 
%let textformat_center = font=Arial height = 12 pt justify = c; 

 
ods escapechar = '^'; 

 
title1 &textformat_left. 'Sample Report Header 1'; 
title2 &textformat_left. 'Sample Report Header 2'; 

 
footnote1 &textformat_left. 'Sample Report Footer 1'; 
footnote2 &textformat_left. 'Sample Report Footer 2'; 

 
ods rtf file = "U:\SGF08\Output\SGF08.rtf" startpage=no notoc_data; 

       /* Create the table */ 
      proc print data=sashelp.class noobs; 
     run; 
 
     quit; 

ods rtf close; 
 
 
Adding the Table of Contents Code 
 
The ODS RTF TEXT= option is used to create the cell in which the RTF control words for creating the actual table 
of contents are imbedded.  The control word \TOC indicates that a table of contents should be created.  The 
control word \f indicates that text marked with the \tc control word will be included as an entry in the TOC.   The 
control word \h indicates that the entry should be created as a hyperlink. 
    ods rtf text = 

     "^S={outputwidth=100% just=l}{\field{\*\fldinst {\\TOC \\f \\h}}}";  
 
The ODS RTF TEXT= option is also used to create the cell in which the RTF control words for marking the TOC 
entries are imbedded.  The control word \tc indicates that the entry is marked for inclusion into the TOC. The 
formatting control word \f3 indicates the font.   The formatting control words \fs0 and \cf8 indicate that the font size 
is 0 and the font color is white.  With a font size of 0 and a font color of white (assuming the report will be printed 
on white paper) the imbedded text becomes “invisible”.    
    ods rtf text =  
          "^S={outputwidth=100% just=l} {\tc\f3\fs0\cf8 Table 1 - Class List}"; 
 
The following is updated code to create the RTF file with one table and one TOC entry: 

ods rtf file = "U:\SGF08\Output\SGF08.rtf" startpage=no notoc_data; 
 
      title3 &textformat_center. bold 'Table of Contents'; 
     ods rtf startpage=now; 

      ods rtf text = 
     "^S={outputwidth=100% just=l}{\field{\*\fldinst {\\TOC \\f \\h}}}";  

 
     /* **************************************** */ 
      /* Page 2 of RTF File */ 
      /* ****************** */ 
 
      /* Start a new page */ 
      title3; 
      ods rtf startpage=now; 
 
      /* Imbed the TOC mark */ 
      ods rtf text =  
          "^S={outputwidth=100% just=l} {\tc\f3\fs0\cf8 Table 1 - Class List}"; 
 
      /* Create the table */ 
      proc print data=sashelp.class noobs; 
     run; 
 
    quit; 

ods rtf close; 
 
 
Rendering the Table of Contents in Microsoft Word 
 
When the RTF file has been created by SAS and opened in Word, you will not immediately see the TOC.  To 
render the TOC press <Ctrl><A> to select all and then press F9 to create the TOC.  Figure 1 displays the TOC 
created with the above code.  Since the TOC entries were specified as hyperlinks, you can press CTRL + click to 
follow the link to the appropriate location in the document.     
 
Figure 1:  Table of Contents with One Entry  
 

 
 
Adding Custom Titles and Footnotes to Tables 
 
By adding additional ODS RTF TEXT= statements, custom titles and footnotes can be added to describe a table.  
Adding the following code to the previous example will create a title and a footnote for Table 1. 
 
    /* **************************************** */ 
    /* Page 2 of RTF File */ 
    /* ****************** */ 
    /* Start a new page */ 
    title3; 
    ods rtf startpage=now; 
 
    /* Print the table title */ 
    ods rtf text=  

   "^S={outputwidth=100% just=l} {\f3\fs24 Table 1 - Class List*}"; 
 
    /* Imbed the TOC mark */ 



Paper PO01 

 5

    ods rtf text=  
   "^S={outputwidth=100% just=l} {\tc\f3\fs0\cf8 Table 1 - Class List}"; 

 
    /* Create the table */ 
    proc print data=sashelp.class noobs; 
    run; 
 
    /* Print the custom footnotes */ 
    ods rtf text=  
       "^S={outputwidth=100% just=l} {\f3\fs24 *This is custom footnote for Table 1}"; 
 
 
Marking Additional Entries for TOC 
 
The following code marks the table of contents entries for tables 2 and 3, both of which will be on the page 3 of 
the RTF document.  The code also adds custom titles and footnotes to these tables.  Figure 2 displays page 3 of 
the RTF document.  Figure 3 displays the rendered table of contents in Microsoft Word after running the 
complete code found in Appendix I. 
 
    /* **************************************** */ 
    /* Page 3 of RTF File */ 
    /* ****************** */ 
    /* Start a new page */ 
    ods rtf startpage=now; 
 
    /* Print the table title */ 
    ods rtf text=  
         "^S={outputwidth=100% just=l} {\line\line\f3\fs24 Table 2 - Gender*}"; 
 
    /* Imbed the TOC mark */ 
    ods rtf text=  

   "^S={outputwidth=100% just=l} {\tc\f3\fs0\cf8 Table 2 - Gender}"; 
 
    /* Create the table */ 
    proc freq data = sashelp.class; 
         table sex; 
    run; 
 
    /* Print the custom footnotes */ 
    ods rtf text=  
         "^S={outputwidth=100% just=l}{\f3\fs24 *This is custom footnote for Table 2}"; 
 
    /* ******************** */ 
    /* Second Table on Page 3 */ 
 
    /* Print the table title */ 
    ods rtf text= "^S={outputwidth=100% just=l}  
         {\line\line\line\line\line\line\f3\fs24 Table 3 - Mean Age†}"; 
 
    /* Imbed the TOC mark */ 
    ods rtf text= "^S={outputwidth=100% just=l} {\tc\f3\fs0\cf8 Table 3 - Mean Age}"; 
 
    /* Create the table */ 
    proc means data = sashelp.class; 
         var age; 
    run; 
 
    /* Print the custom footnotes */ 
    ods rtf text=  
        "^S={outputwidth=100% just=l}{\f3\fs24 †This is custom footnote for Table 3}"; 

Figure 2:  Page 3 of RTF Document 
 

 
 
 
 
Figure 3:  Table of Contents for all Entries 
 

 
 
 
CONCLUSIONS 
 
The technique described in this paper can be used to create an RTF file that contains a custom TOC when 
viewed in Microsoft Word.  The technique involves imbedding “invisible” RTF control words to mark TOC entries 
and create the actual TOC.  Since the TOC is created with RTF code, it is reproducible whenever the RTF file is 
recreated.  With one exception, you have total control over the text of the TOC entries and the placement of the 

TOC marks.  The exception is the inability place a TOC mark in the Header or Footer of a Microsoft Word 
document.  In addition, this paper also describes how to include the results of more than one SAS procedure per 
page, and use the ODS RTF TEXT= statement to add custom titles and footnotes to the results of each 
procedure.     

APPENDIX  I:  Complete Code for Example RTF File 
 
/* APPENDIX  I:  Complete Code for Example RTF File */ 
 
title;  
footnote; 
 
options orientation=landscape nodate nocenter pageno=1; 
 
%let textformat_left   = font=Arial height = 12 pt justify = l; 
%let textformat_center = font=Arial height = 12 pt justify = c; 
 
ods escapechar = '^'; 
 
title1 &textformat_left. 'Sample Report Header 1'; 
title2 &textformat_left. 'Sample Report Header 2'; 
 
footnote1 &textformat_left. 'Sample Report Footer 1'; 
footnote2 &textformat_left. 'Sample Report Footer 2'; 
 
ods rtf file = "U:\SGF08\Output\SGF08.rtf" startpage=no notoc_data; 
 
   /* **************************************** */ 
   /* Page 1 of RTF File */ 
   /* ****************** */ 
 
   /* ******************************************************* */ 
   /* Create Table of Contents        */ 
   /* NOTE:  Once RTF document is open in WORD,     */ 
   /*        press <ctrl>A to select all and F9 to create TOC */ 
   title3 &textformat_center. bold 'Table of Contents'; 
   ods rtf startpage=now; 
   ods rtf text= "^S={outputwidth=100% just=l}{\field{\*\fldinst {\\TOC \\f \\h} } }";  
 
   /* **************************************** */ 
   /* Page 2 of RTF File */ 
   /* ****************** */ 
 
   /* Start a new page */ 
   title3; 
   ods rtf startpage=now; 
 
   /* Print the table title */ 
   ods rtf text= "^S={outputwidth=100% just=l} {\f3\fs24 Table 1 - Class List*}"; 
 
   /* Imbed the TOC mark */ 
   ods rtf text= "^S={outputwidth=100% just=l} {\tc\f3\fs0\cf8 Table 1 - Class List}"; 
 
   /* Create the table */ 
   proc print data=sashelp.class noobs; 
   run; 
 
   /* Print the custom footnotes */ 
   ods rtf text=  
    "^S={outputwidth=100% just=l} {\f3\fs24 *This is custom footnote for Table 1}"; 
 
   /* **************************************** */ 
   /* Page 3 of RTF File */ 
   /* ****************** */ 

   /* Start a new page */ 
   ods rtf startpage=now; 
 
   /* Print the table title */ 
   ods rtf text=  
      "^S={outputwidth=100% just=l} {\line\line\f3\fs24 Table 2 - Gender*}"; 
 
   /* Imbed the TOC mark */ 
   ods rtf text= "^S={outputwidth=100% just=l} {\tc\f3\fs0\cf8 Table 2 - Gender}"; 
 
   /* Create the table */ 
   proc freq data = sashelp.class; 
      table sex; 
   run; 
 
   /* Print the custom footnotes */ 
   ods rtf text=  
      "^S={outputwidth=100% just=l} {\f3\fs24 *This is custom footnote for Table 2}"; 
 
   /* ******************** */ 
   /* Second Table on Page 3 */ 
 
   /* Print the table title */ 
   ods rtf text= "^S={outputwidth=100% just=l}  
      {\line\line\line\line\line\line\f3\fs24 Table 3 - Mean Age†}"; 
 
   /* Imbed the TOC mark */ 
   ods rtf text= "^S={outputwidth=100% just=l} {\tc\f3\fs0\cf8 Table 3 - Mean Age}"; 
 
   /* Create the table */ 
   proc means data = sashelp.class; 
      var age; 
   run; 
 
   /* Print the custom footnotes */ 
   ods rtf text=  
    "^S={outputwidth=100% just=l} {\f3\fs24 †This is custom footnote for Table 3}"; 
 
 
   /* **************************************** */ 
   /* Page 4 of RTF File */ 
   /* Start a new page */ 
   ods rtf startpage=now; 
 
   /* Print the table title */ 
   ods rtf text= "^S={outputwidth=100% just=l} {\f3\fs24 Table 4 - Mean Weight*}" ; 
 
   /* Imbed the TOC mark */ 
   ods rtf text=  
     "^S={outputwidth=100% just=l} {\tc\f3\fs0\cf8 Table 4 - Mean Weight}"; 
 
   /* Create the table */ 
   proc means data = sashelp.class; 
      var weight; 
   run; 
 
   /* Print the custom footnotes */ 
   ods rtf text=  
    "^S={outputwidth=100% just=l} {\f3\fs24 *This is custom footnote for Table 4}"; 

   /* ******************** */ 
   /* Second Table on Page 4 */ 
 
   /* Print the table title */ 
   ods rtf text=  
   "^S={outputwidth=100% just=l}{\line\line\line\line\f3\fs24 Table 5 - Mean Height†}" ; 
 
   /* Imbed the TOC mark */ 
   ods rtf text= "^S={outputwidth=100% just=l} {\tc\f3\fs0\cf8 Table 5 - Mean Height}"; 
 
   /* Create the table */ 
   proc means data = sashelp.class; 
      var Height; 
   run; 
 
   /* Print the custom footnotes */ 
   ods rtf text=  
    "^S={outputwidth=100% just=l} {\f3\fs24 †This is custom footnote for Table 5}"; 
 
 
   quit; 
 
ods rtf close; 


